CME LIVE OPEN ACCESS

Electrical Electronics Engineering: Open Access

Research Article

Normalization of Densities of Distributions of Output Processes of Dynamic Systems using Hermite Polynomial Decomposition in the Environment of the Maple

G Gasimov*, E Rzayev and M Aghayeva

Baku State University, Azerbaijan.

Corresponding Author: G. Gasimov, Baku State University, Azerbaijan.

Received: 2025 Dec 29 **Accepted:** 2026 Jan 16 **Published:** 2026 Jan 26

In the environment of the Maple mathematical package, approximate expressions obtained in the form of decomposition by orthogonal Hermite polynomials for the densities of distributions of output processes of dynamical systems determined by integral operators with weights of "unit ", "exponential" and "exponential - sinusoidal" types and with input signals in the form of a square of a normal stationary random function are investigated [1-3].

The program was compiled based on the following heuristic algorithm:
Step 1) input of the correlation function of the input signal and the weight functions;
Step 2) calculation of the correlation matrix to determine the second moments of the output signals through the characteristic function of the 4 - dimensional normal system generated by the input signal;
Step 3) calculation of the correlation matrix to determine the third moments of the output signals through the characteristic function of the 6 - dimensional normal system;
Step 4) determination of the fourth moments through an 8 - dimensional normal system;
Step 5) calculation of asymmetry and kurtosis for the output process of each dynamic system;
Step 6) definition of expressions for densities based on the calculated characteristics; step7) construction of comparative density graphs by increasing time values.

The obtained figures allow us to draw practically useful: for the first two types of weight functions, an almost normal distribution is obtained for sufficiently large time values, but for the third type of weight functions, the difference from the normal distribution law is quite significant.

restart;
In the Maple, approximate expressions obtained in the form of decomposition by orthogonal Hermite polynomials

$$H_l(x) = (-1)^l e^{-x^2} \frac{d^l}{dx^l}\left(e^{-x^2}\right), \quad l = 1, 2, \ldots$$

for the densities of distributions of output processes of dynamical systems determined
by integral operators with weights of "unit", "exponential" and "exponential - sinusoidal" types

$$\mathrm{Y(t)} = \int_0^t X(\tau)\, d\tau \;; \quad Y(t) = \int_0^t e^{-\rho(t-\tau)} X(\tau)\, d\tau;$$

$$Y(t) = \int_0^t \frac{h_0^2}{\omega} e^{-h_1(t-\tau)} \sin\omega(t-\tau) X(\tau)\, d\tau \;; \rho > 0\,, 0 < h_1 < h_0,\ \omega = \sqrt{h_0^2 - h_1^2}$$

and with input signals in the form of a square of a normal stationary random function (X(t) = Z(t)^2, Z(t) ∈ N(0, s^2)) are investigated.

An approximate decomposition of the distribution density over Hermite polynomials is known:

$$g_Y(y) = \frac{1}{\sigma_y}\varphi\left(\frac{y-\bar{y}}{\sigma_y}\right) + \frac{1}{3!}\frac{Sk(Y)}{\sigma_y}\varphi^{(3)}\left(\frac{y-\bar{y}}{\sigma_y}\right) + \frac{1}{4!}\frac{Ex(Y)}{\sigma_y}\varphi^{(4)}\left(\frac{y-\bar{y}}{\sigma_y}\right), (H)$$

where ϕ(x) is the density of the standard normal distribution, Sk(Y) and Ex(Y) are the "asymmetry" and "kurtosis", respectively, of the distribution law Y(t):

$$\mathrm{Sk}(Y) = \frac{\mu_3(Y)}{{}_y\sigma^3}, \quad Ex(Y) = \frac{\mu_4(Y)}{\sigma_y^4} - 3,$$

$\mu_n(Y)$ is the nth central moment of Y(t).
Using the formula

$$m_1(Y) = \int_0^t l(t-\tau)\,\bar{x}(\tau)\,d\tau,$$

we determine the mathematical expectation Y(t) (m1 , m1r , m1s) for each of the following cases:

$$\mathrm{l}(t) \equiv 1, \quad \mathrm{l}(t) = \mathrm{e}^{-\rho t} \quad \mathrm{l}(t) = \frac{h_0^2}{\omega}\,\mathrm{e}^{-th_1} sin\omega t \left(\omega = \sqrt{h_0^2 - h_1^2}\right);$$

in *the program*, *the last weight function participates under the* identifier ls .

```
> assume(t>0,a>0,s>0,rho>0,h0>h1);
> Kz:=unapply(s^2*exp(-a*abs(t2-t1)),t1,t2,s,a);
```

$$Kz := (t1, t2, s\sim, a\sim) \mapsto s\sim^2 \cdot \mathrm{e}^{-a\sim\cdot|t1-t2|} \quad (1)$$

```
> m1:=unapply(s^2*t,t,s);
```

$$m1 := (t\sim, s\sim) \mapsto s\sim^2 \cdot t\sim \quad (2)$$

```
> l:=unapply(exp(-rho*t),t,rho);
```

$$l := (t\sim, \rho\sim) \mapsto \mathrm{e}^{-t\sim\cdot\rho\sim} \quad (3)$$

```
> m1r:=unapply((s^2/rho)*(1-exp(-rho*t)),t,s,rho);
```

$$m1r := (t\sim, s\sim, \rho\sim) \mapsto \frac{s\sim^2 \cdot \left(1 - \mathrm{e}^{-\rho\sim\cdot t\sim}\right)}{\rho\sim} \quad (4)$$

```
> omega:=unapply(sqrt(h0^2-h1^2),h0,h1);
```

$$\omega := (h0\sim, h1\sim) \mapsto \sqrt{h0\sim^2 - h1\sim^2} \quad (5)$$

```
> ls:=unapply((h0^2/omega(h0,h1))*exp(-h1*t)*sin(omega(h0,h1)*
  t),t,h0,h1);
```

$$ls := (t\sim, h0\sim, h1\sim) \mapsto \frac{h0\sim^2 \cdot \mathrm{e}^{-h1\sim\cdot t\sim} \cdot \sin\left(\sqrt{h0\sim^2 - h1\sim^2} \cdot t\sim\right)}{\sqrt{h0\sim^2 - h1\sim^2}} \quad (6)$$

```
> m1s:=unapply(s^2*int(ls(t-t1,h0,h1),t1=0..t),t,s,h0,h1);
```

$$m1s := (t\sim, s\sim, h0\sim, h1\sim) \mapsto -\frac{1}{\sqrt{h0\sim^2 - h1\sim^2}} \Big(s\sim^2 \cdot \Big(\mathrm{e}^{-h1\sim \cdot t\sim} \cdot \cos\big(\sqrt{h0\sim^2 - h1\sim^2} \cdot t\sim\big) \cdot \sqrt{h0\sim^2 - h1\sim^2} + \mathrm{e}^{-h1\sim \cdot t\sim} \cdot \sin\big(\sqrt{h0\sim^2 - h1\sim^2} \cdot t\sim\big) \cdot h1\sim - \sqrt{h0\sim^2 - h1\sim^2}\Big)\Big) \quad (7)$$

We introduce a characteristic function for a 4-dimensional normal system (under the identifier G), and the elements of the b_{ij} corresponding correlation matrix are calculated using K_z .

```
> w:=array(1..4):b:=array(1..4,1..4):
> G:=unapply(exp(-(1/2)*sum(sum(b[i,j]*w[i]*w[j],i=1..4),j=1..4)),b,w);
```

$$G := (b, w) \mapsto \mathrm{e}^{-\frac{1}{2} \cdot b_{1,1} \cdot w_1^2 - \frac{1}{2} \cdot b_{1,2} \cdot w_1 \cdot w_2 - \frac{1}{2} \cdot b_{1,3} \cdot w_1 \cdot w_3 - \frac{1}{2} \cdot b_{1,4} \cdot w_1 \cdot w_4 - \frac{1}{2} \cdot b_{2,1} \cdot w_1 \cdot w_2 - \frac{1}{2} \cdot b_{2,2} \cdot w_2^2 - \frac{1}{2} \cdot b_{2,3} \cdot w_2 \cdot w_3 - \frac{1}{2} \cdot b_{2,4} \cdot w_2 \cdot w_4 - \frac{1}{2} \cdot b_{3,1} \cdot w_1 \cdot w_3 - \frac{1}{2} \cdot b_{3,2} \cdot w_2 \cdot w_3 - \frac{1}{2} \cdot b_{3,3} \cdot w_3^2 - \frac{1}{2} \cdot b_{3,4} \cdot w_3 \cdot w_4 - \frac{1}{2} \cdot b_{4,1} \cdot w_1 \cdot w_4 - \frac{1}{2} \cdot b_{4,2} \cdot w_2 \cdot w_4 - \frac{1}{2} \cdot b_{4,3} \cdot w_3 \cdot w_4 - \frac{1}{2} \cdot b_{4,4} \cdot w_4^2} \quad (8)$$

```
> for j from 1 to 4 do b[j,j]:=b[1,1] od:
> DG:=unapply(diff(G(b,w),w[1],w[2],w[3],w[4]),b,w):
```

Using the known properties of the characteristic function, we obtain:

$$\mu_2\ (X) = \frac{1}{i^4} \cdot \frac{\partial^4}{\partial w_1\, \partial w_2 \partial w_3 \partial w_4} G(w_1, w_2, w_3, w_4)\Big|_{w_1 = w_2 = w_3 = w_4 = 0}$$

(under the identifier mu2)) .

```
> DG0:=unapply(subs({w[1]=0,w[2]=0,w[3]=0,w[4]=0},DG(b,w)),b):
> mu2x:=unapply(simplify(subs({b[1,2]=b[1,1],b[2,1]=b[1,1],b[3,4]=b[1,1],b[4,3]=b[1,1],

                              b[3,1]=b[1,3],b[1,4]=b[1,3],b[4,1]=b[1,3],b[2,3]=b[1,3],b[3,2]=b[1,3],b[2,4]=b[1,3],b[4,2]=b[1,3]},

               DG0(b))),b);
```

$$mu2x := b \mapsto b_{1,1}^2 + 2 \cdot b_{1,3}^2 \quad (9)$$

```
> b[1,1]:=s^2:
> b[1,3]:=Kz(t1,t2,s,a):
> mu2:=unapply(mu2x(b),t1,t2,s,a):
```

Squaring both parts of the relation

$$\mathrm{Y(t)} = \int_0^t l(t-\tau) X(\tau)\ \mathrm{d}\tau \qquad (L)$$

and representing the square of the last integral , after finding the mathematical expectation, we get:

Citation: Gasimov, G., Rzayev, E., Aghayeva, M. (2026). Normalization of Densities of Distributions of Output Processes of Dynamic Systems using Hermite Polynomial Decomposition in the Environment of the Maple. *Electr Electron Eng Open Access, 3*(1), 1-16.

$$m_2(Y) = \int_0^t \int_0^t l(t-t_1)\, l(t-t_2) mu2(t_1, t_2, s, a)\, dt_1\, dt_2 \quad (L2)$$

According to the formula (L2) , the initial moments of the 2nd order under the identifier m2, m2r, m2s are calculated, respectively.

```
> m2:=unapply(int(int(mu2(t1,t2,s,a),t1=0..t),t2=0..t),t,a,s);
```

$$m2 := (t\sim, a\sim, s\sim) \mapsto \frac{s\sim^4 \cdot \left(t\sim^2 \cdot a\sim^2 + 2\cdot t\sim \cdot a\sim + e^{-2\cdot t\sim\cdot a\sim} - 1\right)}{a\sim^2} \quad \textbf{(10)}$$

```
> m2r:=unapply(int(int(l(t-t1,rho)*l(t-t2,rho)*mu2(t1,t2,s,a),
  t1=0..t),t2=0..t),t,a,s,rho);
```

$$m2r := (t\sim, a\sim, s\sim, \rho\sim) \mapsto \frac{1}{(2\cdot a\sim + \rho\sim)\cdot(2\cdot a\sim - \rho\sim)\cdot \rho\sim^2} \Big(\big(4\cdot \rho\sim^2 \cdot e^{-t\sim\cdot(2\cdot a\sim + \rho\sim)} - 8\cdot a\sim^2 \cdot e^{-\rho\sim\cdot t\sim} + 2\cdot \rho\sim^2 \cdot e^{-\rho\sim \cdot t\sim} + 4\cdot a\sim^2 \cdot e^{-2\cdot\rho\sim\cdot t\sim} - 4\cdot a\sim \cdot e^{-2\cdot \rho\sim \cdot t\sim}\cdot \rho\sim - 3\cdot \rho\sim^2 \cdot e^{-2\cdot\rho\sim\cdot t\sim} + 4\cdot a\sim^2 + 4\cdot a\sim \cdot \rho\sim - 3\cdot \rho\sim^2 \big)\cdot s\sim^4 \Big) \quad \textbf{(11)}$$

```
> m2s:=unapply(int(int(ls(t-t1,h0,h1)*ls(t-t2,h0,h1)*mu2(t1,
  t2,s,a),t1=0..t),t2=0..t),t,a,s,h0,h1):
```

The central moments of the 2nd odrer are calculated:

```
> mu2y:=unapply(simplify(m2(t,a,s)-m1(t,s)^2),t,a,s);
```

$$mu2y := (t\sim, a\sim, s\sim) \mapsto \frac{s\sim^4 \cdot \left(2\cdot t\sim \cdot a\sim + e^{-2\cdot t\sim \cdot a\sim} - 1\right)}{a\sim^2} \quad \textbf{(12)}$$

```
> mu2yr:=unapply(simplify(m2r(t,a,s,rho)-m1r(t,s,rho)^2),t,a,
  s,rho);
```

$$mu2yr := (t\sim, a\sim, s\sim, \rho\sim) \mapsto \quad \textbf{(13)}$$

$$-\frac{4\cdot s\sim^4 \cdot \left(-e^{-t\sim\cdot(2\cdot a\sim + \rho\sim)}\cdot \rho\sim + \left(a\sim + \frac{\rho\sim}{2}\right)\cdot e^{-2\cdot\rho\sim\cdot t\sim} - a\sim + \frac{\rho\sim}{2}\right)}{4\cdot a\sim^2 \cdot \rho\sim - \rho\sim^3}$$

```
> mu2ys:=unapply(simplify(m2s(t,a,s,h0,h1)-m1s(t,s,h0,h1)^2),
  t,a,s,h0,h1);
```

$$mu2ys := (t\sim, a\sim, s\sim, h0\sim, h1\sim) \mapsto \Bigg(16\cdot h0\sim^2 \cdot \bigg(e^{-2\cdot h1\sim \cdot t\sim}\cdot \left(a\sim^2 + a\sim \cdot h1\sim + \frac{1}{4} \cdot h0\sim^2\right)\cdot h1\sim \cdot \sqrt{h0\sim^2 - h1\sim^2} \cdot \left(a\sim \cdot h1\sim + \frac{1}{2}\cdot h0\sim^2 - h1\sim^2\right) \cdot \cos\left(\sqrt{h0\sim^2 - h1\sim^2}\cdot t\sim\right)^2 - h1\sim \cdot (h0\sim + h1\sim)\cdot(h0\sim - h1\sim)\cdot \Big((-h1\sim + a\sim) \cdot e^{-2\cdot h1\sim\cdot t\sim}\cdot \left(a\sim^2 + a\sim\cdot h1\sim + \frac{1}{4}\cdot h0\sim^2\right)\cdot \sin\left(\sqrt{h0\sim^2 - h1\sim^2}\cdot t\sim\right) \quad \textbf{(14)}$$

$$+\frac{h0\sim^2\cdot\sqrt{h0\sim^2-h1\sim^2}\cdot e^{-t\sim\cdot(h1\sim+2\cdot a\sim)}}{4}\Bigg)\cdot\cos\left(\sqrt{h0\sim^2-h1\sim^2}\cdot t\sim\right)$$

$$-\frac{1}{2}\left(h0\sim^2\cdot e^{-t\sim\cdot(h1\sim+2\cdot a\sim)}\cdot h1\sim\cdot(h0\sim+h1\sim)\cdot\left(a\sim+\frac{h1\sim}{2}\right)\cdot(h0\sim-h1\sim)\right.$$
$$\left.\cdot\sin\left(\sqrt{h0\sim^2-h1\sim^2}\cdot t\sim\right)\right)-\frac{1}{2}\left(\sqrt{h0\sim^2-h1\sim^2}\cdot\left(\left(a\sim^2+a\sim\cdot h1\sim+\frac{1}{4}\right.\right.\right.$$
$$\left.\cdot h0\sim^2\right)\cdot\left(a\sim\cdot h0\sim^2+a\sim\cdot h1\sim^2-h1\sim^3\right)\cdot e^{-2\cdot h1\sim\cdot t\sim}-(a\sim+h1\sim)\cdot\left(a\sim^2-a\sim\cdot h1\sim\right.$$
$$\left.\left.\left.\left.+\frac{1}{4}\cdot h0\sim^2\right)\cdot(h0\sim+h1\sim)\cdot(h0\sim-h1\sim)\right)\right)\right)\cdot s\sim^4\Bigg)\Bigg/\left(\left(h0\sim^2-h1\sim^2\right)^{3/2}\cdot\left(\right.\right.$$
$$\left.\left.-16\cdot a\sim^2\cdot h1\sim^3+16\cdot\left(a\sim^2+\frac{h0\sim^2}{4}\right)^2\cdot h1\sim\right)\right)$$

We introduce a characteristic function for a 6-dimensional normal system (under the identifier F), and the elements c_{ij} of the corresponding correlation matrix are calculated via Kz:

```
> v:=array(1..6):c:=array(1..6,1..6):
> F:=unapply(exp(-(1/2)*sum(sum(c[i,j1]*v[i]*v[j1],i=1..6),j1=
  1..6)),c,v):
> for j from 1 to 6 do c[j,j]:=c[1,1] od:
> DF:=unapply(diff(F(c,v),v[1],v[2],v[3],v[4],v[5],v[6]),c,v):
```

Using F(c,v), we obtain:

$$\mu_3(X)=\frac{1}{i^6}\frac{\partial^6}{\partial v_1\,\partial v_2\,\partial v_3\,\partial v_4\,\partial v_5\,\partial v_6}F(c,v)\Bigg|_{v_1=v_2=v_3=v_4=v_5=v_6=0}.$$

```
> DF0:=unapply(subs({v[1]=0,v[2]=0,v[3]=0,v[4]=0,v[5]=0,v[6]=
  0},DF(c,v)),c):
> mu3x:=unapply(simplify(-subs({c[1,2]=c[1,1],c[2,1]=c[1,1],c
  [3,4]=c[1,1],c[4,3]=c[1,1],c[5,6]=c[1,1],c[6,5]=c[1,1],

                              c[3,1]=c[1,3],c[1,4]=c[1,3],c[4,1]=c[1,
  3],c[2,3]=c[1,3],c[3,2]=c[1,3],c[2,4]=c[1,3],c[4,2]=c[1,3],

                c[5,1]=c[1,5],c[1,6]=c[1,5],c[6,1]=c[1,5],c[2,5]=
  c[1,5],c[5,2]=c[1,5],c[2,6]=c[1,5],c[6,2]=c[1,5],

   c[5,3]=c[3,5],c[4,5]=c[3,5],c[5,4]=c[3,5],c[3,6]=c[3,5],c
  [6,3]=c[3,5],c[4,6]=c[3,5],c[6,4]=c[3,5]},DF0(c))),c);
```

$$mu3x:=c\mapsto c_{1,1}^3+\left(2\cdot c_{1,3}^2+2\cdot c_{1,5}^2+2\cdot c_{3,5}^2\right)\cdot c_{1,1}+8\cdot c_{1,3}\cdot c_{1,5}\cdot c_{3,5}\qquad(15)$$

```
> c[1,1]:=s^2:
> c[1,3]:=Kz(t1,t2,s,a):
> c[1,5]:=Kz(t1,t3,s,a):
> c[3,5]:=Kz(t2,t3,s,a):
> mu3:=unapply(mu3x(c),t1,t2,t3,s,a):
```

We raise both parts of the relation (L) to the 3rd degree and represent the cube of the right side as a triple integral, after calculating the mahematical expectaton we get:

$$m_3(Y)=\int_0^t\int_0^t\int_0^t l(t-t_1)\,l(t-t_2)\,l(t-t_3)\,mu3(t_1,t_2,t_3,s,a)\,\mathrm{d}t_1\,\mathrm{d}t_2\,\mathrm{d}t_3. \qquad (L3)$$

According to the formula (L3) , the initial moments of the 3rd order under the identifier m3, m3r and m3sn are calculated (for a quick calculation of m3sn, we perform fragmentation according to the formula (15)) .

```
> m3:=unapply(int(int(int(mu3(t1,t2,t3,s,a),t1=0..t),t2=0..t),
  t3=0..t),t,a,s);
```

$$m3 := (t\sim, a\sim, s\sim) \mapsto \frac{1}{a\sim^3}\left(s\sim^6\cdot\left(t\sim^3\cdot a\sim^3+6\cdot t\sim^2\cdot a\sim^2+15\cdot t\sim\cdot \mathrm{e}^{-2\cdot t\sim\cdot a\sim}\cdot a\sim+9\cdot t\sim\cdot a\sim+12\cdot \mathrm{e}^{-2\cdot t\sim\cdot a\sim}-12\right)\right) \qquad (16)$$

```
> m3r:=unapply(int(int(int(l(t-t1,rho)*l(t-t2,rho)*l(t-t3,rho)
  *mu3(t1,t2,t3,s,a),t1=0..t),t2=0..t),t3=0..t),t,a,s,rho);
```

$$m3r := (t\sim, a\sim, s\sim, \rho\sim) \mapsto \frac{1}{\rho\sim^3\cdot\left(4\cdot a\sim^4-5\cdot a\sim^2\cdot\rho\sim^2+\rho\sim^4\right)}\Big(s\sim^6\cdot\Big(12\cdot a\sim^4\cdot \mathrm{e}^{-2\cdot\rho\sim\cdot t\sim}+9\cdot \mathrm{e}^{-2\cdot\rho\sim\cdot t\sim}\cdot\rho\sim^4-4\cdot a\sim^4\cdot \mathrm{e}^{-3\cdot\rho\sim\cdot t\sim}-15\cdot \mathrm{e}^{-3\cdot\rho\sim\cdot t\sim}\cdot\rho\sim^4-12\cdot a\sim^4\cdot \mathrm{e}^{-\rho\sim\cdot t\sim}-9\cdot \mathrm{e}^{-\rho\sim\cdot t\sim}\cdot\rho\sim^4-36\cdot\rho\sim^4\cdot \mathrm{e}^{-2\cdot t\sim\cdot a\sim-\rho\sim\cdot t\sim}+24\cdot\rho\sim^4\cdot \mathrm{e}^{-2\cdot t\sim\cdot a\sim-2\cdot\rho\sim\cdot t\sim}+12\cdot\rho\sim^4\cdot \mathrm{e}^{-2\cdot t\sim\cdot(a\sim+\rho\sim)}+12\cdot a\sim^3\cdot\rho\sim-36\cdot a\sim\cdot\rho\sim^3+4\cdot a\sim^4+15\cdot\rho\sim^4-21\cdot \mathrm{e}^{-2\cdot\rho\sim\cdot t\sim}\cdot a\sim^2\cdot\rho\sim^2+12\cdot \mathrm{e}^{-3\cdot\rho\sim\cdot t\sim}\cdot a\sim^3\cdot\rho\sim-5\cdot \mathrm{e}^{-3\cdot\rho\sim\cdot t\sim}\cdot a\sim^2\cdot\rho\sim^2-36\cdot \mathrm{e}^{-3\cdot\rho\sim\cdot t\sim}\cdot a\sim\cdot\rho\sim^3-12\cdot \mathrm{e}^{-\rho\sim\cdot t\sim}\cdot a\sim^3\cdot\rho\sim+21\cdot \mathrm{e}^{-\rho\sim\cdot t\sim}\cdot a\sim^2\cdot\rho\sim^2+12\cdot \mathrm{e}^{-\rho\sim\cdot t\sim}\cdot a\sim\cdot\rho\sim^3-12\cdot \mathrm{e}^{-2\cdot\rho\sim\cdot t\sim}\cdot a\sim^3\cdot\rho\sim+12\cdot \mathrm{e}^{-2\cdot\rho\sim\cdot t\sim}\cdot a\sim\cdot\rho\sim^3+60\cdot a\sim^2\cdot\rho\sim^2\cdot \mathrm{e}^{-2\cdot t\sim\cdot a\sim-\rho\sim\cdot t\sim}+24\cdot a\sim\cdot\rho\sim^3\cdot \mathrm{e}^{-2\cdot t\sim\cdot a\sim-\rho\sim\cdot t\sim}-40\cdot a\sim^2\cdot\rho\sim^2\cdot \mathrm{e}^{-2\cdot t\sim\cdot a\sim-2\cdot\rho\sim\cdot t\sim}+16\cdot a\sim\cdot\rho\sim^3\cdot \mathrm{e}^{-2\cdot t\sim\cdot a\sim-2\cdot\rho\sim\cdot t\sim}-20\cdot a\sim^2\cdot\rho\sim^2\cdot \mathrm{e}^{-2\cdot t\sim\cdot(a\sim+\rho\sim)}+8\cdot a\sim\cdot\rho\sim^3\cdot \mathrm{e}^{-2\cdot t\sim\cdot(a\sim+\rho\sim)}+5\cdot a\sim^2\cdot\rho\sim^2\Big)\Big) \qquad (17)$$

```
> m3s0:=unapply(int(int(ls(t-t1,h0,h1)*ls(t-t2,h0,h1)*s^6,t1=
  0..t),t2=0..t),t,s,h0,h1):
> m3s1:=unapply(int(int(ls(t-t1,h0,h1)*ls(t-t2,h0,h1)*2*s^2*Kz
  (t1,t2,s,a)^2,t1=0..t),t2=0..t),t,a,s,h0,h1):
> m3s2:=unapply(int(int(ls(t-t1,h0,h1)*ls(t-t2,h0,h1)*2*s^2*Kz
  (t1,t3,s,a)^2,t1=0..t),t2=0..t),t,t3,a,s,h0,h1):
> m3s3:=unapply(int(int(ls(t-t1,h0,h1)*ls(t-t2,h0,h1)*2*s^2*Kz
  (t2,t3,s,a)^2,t1=0..t),t2=0..t),t,t3,a,s,h0,h1):
> m3s4:=unapply(int(int(ls(t-t1,h0,h1)*ls(t-t2,h0,h1)*8*Kz(t1,
  t2,s,a)*Kz(t1,t3,s,a)*Kz(t2,t3,s,a),t1=0..t),t2=0..t),t,t3,
  a,s,h0,h1):
> m3s:=unapply(m3s0(t,s,h0,h1)+m3s1(t,a,s,h0,h1)+m3s2(t,t3,a,
  s,h0,h1)+m3s3(t,t3,a,s,h0,h1)+m3s4(t,t3,a,s,h0,h1),t,t3,a,s,
  h0,h1):
> m3sn:=unapply(int(ls(t-t3,h0,h1)*m3s(t,t3,a,s,h0,h1),t3=0..
  t),t,a,s,h0,h1):
```

The central moments of the 3rd order are calculated:

```
> mu3y:=unapply(simplify(m3(t,a,s)-3*m1(t,s)*m2(t,a,s)+2*m1(t,s)^3),t,a,s);
```

$$mu3y := (t\sim, a\sim, s\sim) \mapsto \frac{12 \cdot s\sim^6 \cdot \left((t\sim \cdot a\sim + 1) \cdot e^{-2 \cdot t\sim \cdot a\sim} + t\sim \cdot a\sim - 1 \right)}{a\sim^3} \quad (18)$$

```
> mu3yr:=unapply(simplify(m3r(t,a,s,rho)-3*m1r(t,s,rho)*m2r(t,a,s,rho)+2*m1r(t,s,rho)^3),t,a,s,rho);
```

$$mu3yr := (t\sim, a\sim, s\sim, \rho\sim) \mapsto -\frac{1}{4 \cdot a\sim^4 \cdot \rho\sim - 5 \cdot a\sim^2 \cdot \rho\sim^3 + \rho\sim^5} \Bigg(16 \cdot \Bigg(-3 \cdot (a\sim + \rho\sim) \quad (19)$$

$$\cdot \left(a\sim - \frac{\rho\sim}{2} \right) \cdot e^{-t\sim \cdot (2 \cdot a\sim + \rho\sim)} + 3 \cdot \left(a\sim + \frac{\rho\sim}{2} \right) \cdot (a\sim - \rho\sim) \cdot e^{-2 \cdot t\sim \cdot (a\sim + \rho\sim)}$$

$$+ (a\sim + \rho\sim) \cdot \left(a\sim + \frac{\rho\sim}{2} \right) \cdot e^{-3 \cdot \rho\sim \cdot t\sim} - (a\sim - \rho\sim) \cdot \left(a\sim - \frac{\rho\sim}{2} \right) \Bigg) \cdot s\sim^6 \Bigg)$$

```
> mu3ys:=unapply(simplify(m3sn(t,a,s,h0,h1)-3*m1s(t,s,h0,h1)*m2s(t,a,s,h0,h1)+2*m1s(t,s,h0,h1)^3),t,a,s,h0,h1);
```

$$mu3ys := (t\sim, a\sim, s\sim, h0\sim, h1\sim) \mapsto \Bigg(64 \cdot h0\sim^4 \cdot \Bigg(\Bigg(a\sim^2 - 3 \cdot a\sim \cdot h1\sim - \frac{1}{2} \cdot h0\sim^2 + 2 \quad (20)$$

$$\cdot h1\sim^2 \Bigg) \cdot e^{-3 \cdot h1\sim \cdot t\sim} \cdot (a\sim + h1\sim) \cdot (-h1\sim + a\sim) \cdot \sqrt{h0\sim^2 - h1\sim^2} \cdot (h0\sim^2 + 8 \cdot h1\sim^2)$$

$$\cdot \left(a\sim^2 + a\sim \cdot h1\sim + \frac{1}{4} \cdot h0\sim^2 \right) \cdot (a\sim^2 + 2 \cdot a\sim \cdot h1\sim + h0\sim^2) \cdot \cos\left(\sqrt{h0\sim^2 - h1\sim^2} \cdot t\sim \right)^3$$

$$+ (a\sim + h1\sim) \cdot (-h1\sim + a\sim) \cdot (h0\sim^2 + 8 \cdot h1\sim^2) \cdot \left(a\sim^2 + a\sim \cdot h1\sim + \frac{1}{4} \cdot h0\sim^2 \right)$$

$$\cdot \Bigg(\left(a\sim \cdot h1\sim + \frac{3}{2} \cdot h0\sim^2 - 2 \cdot h1\sim^2 \right) \cdot e^{-3 \cdot h1\sim \cdot t\sim} \cdot (-h1\sim + a\sim) \cdot (a\sim^2 + 2 \cdot a\sim \cdot h1\sim$$

$$+ h0\sim^2) \cdot \sin\left(\sqrt{h0\sim^2 - h1\sim^2} \cdot t\sim \right) - \frac{1}{2} \Big(3 \cdot \sqrt{h0\sim^2 - h1\sim^2} \cdot e^{-2 \cdot t\sim \cdot (a\sim + h1\sim)}$$

$$\cdot (a\sim^2 - 2 \cdot a\sim \cdot h1\sim + h0\sim^2) \cdot (a\sim \cdot h1\sim - h0\sim^2 + 2 \cdot h1\sim^2) \Big) \Bigg)$$

$$\cdot \cos\left(\sqrt{h0\sim^2 - h1\sim^2} \cdot t\sim \right)^2 - 3 \cdot (a\sim + h1\sim) \cdot \Bigg(-\frac{1}{2} \Bigg((a\sim^2 - 2 \cdot a\sim \cdot h1\sim$$

$$+ h0\sim^2) \cdot (-h1\sim + a\sim) \cdot (h0\sim - h1\sim) \cdot (h0\sim^2 + 8 \cdot h1\sim^2) \cdot (h0\sim + h1\sim)$$

$$\cdot e^{-2 \cdot t\sim \cdot (a\sim + h1\sim)} \cdot \left(a\sim^2 + a\sim \cdot h1\sim + \frac{1}{4} \cdot h0\sim^2 \right) \cdot (a\sim + 2 \cdot h1\sim) \cdot \sin\Big(\sqrt{h0\sim^2 - h1\sim^2}$$

$$\cdot t\sim)\Big) + \sqrt{h0\sim^2 - h1\sim^2}\cdot\Big(\frac{1}{4}\Big((a\sim - 2\cdot h1\sim)\cdot(h0\sim - h1\sim)\cdot(h0\sim^2+8\cdot h1\sim^2)$$

$$\cdot(h0\sim + h1\sim)\cdot\Big(a\sim^2 - a\sim\cdot h1\sim + \frac{1}{4}\cdot h0\sim^2\Big)\cdot e^{-t\sim\cdot(h1\sim+2\cdot a\sim)}\Big) + e^{-3\cdot h1\sim\cdot t\sim}\cdot\Big(a\sim^2$$

$$+ a\sim\cdot h1\sim + \frac{1}{4}\cdot h0\sim^2\Big)\cdot\Big((h0\sim^2+2\cdot h1\sim^2)\cdot a\sim^3 + (-4\cdot h0\sim^2\cdot h1\sim - 8\cdot h1\sim^3)\cdot a\sim^2$$

$$+\Big(\frac{1}{4}\cdot h0\sim^4 + \frac{13}{4}\cdot h0\sim^2\cdot h1\sim^2 + 10\cdot h1\sim^4\Big)\cdot a\sim - \frac{h0\sim^2\cdot h1\sim^3}{2} - 4\cdot h1\sim^5\Big)\Big)\cdot(a\sim^2$$

$$+2\cdot a\sim\cdot h1\sim + h0\sim^2)\Big)\cdot\cos\Big(\sqrt{h0\sim^2-h1\sim^2}\cdot t\sim\Big) - 7\cdot\Big(\frac{1}{14}\Big(3\cdot\Big(a\sim+\frac{h1\sim}{2}\Big)$$

$$\cdot(a\sim-2\cdot h1\sim)\cdot(h0\sim-h1\sim)\cdot(h0\sim^2+8\cdot h1\sim^2)\cdot(h0\sim+h1\sim)\cdot\Big(a\sim^2 - a\sim\cdot h1\sim + \frac{1}{4}$$

$$\cdot h0\sim^2\Big)\cdot e^{-t\sim\cdot(h1\sim+2\cdot a\sim)}\Big) + \Big(\Big(\frac{2}{7}\cdot h1\sim^3 + h0\sim^2\cdot h1\sim\Big)\cdot a\sim^3 + \Big(\frac{3}{14}\cdot h0\sim^4 - \frac{16}{7}$$

$$\cdot h0\sim^2\cdot h1\sim^2 - \frac{8}{7}\cdot h1\sim^4\Big)\cdot a\sim^2 + \Big(\frac{3}{28}\cdot h0\sim^4\cdot h1\sim + \frac{29}{28}\cdot h0\sim^2\cdot h1\sim^3 + \frac{10}{7}\cdot h1\sim^5\Big)$$

$$\cdot a\sim - \frac{h0\sim^2\cdot h1\sim^4}{14} - \frac{4\cdot h1\sim^6}{7}\Big)\cdot e^{-3\cdot h1\sim\cdot t\sim}\cdot\Big(a\sim^2 + a\sim\cdot h1\sim + \frac{1}{4}\cdot h0\sim^2\Big)\Big)\cdot(a\sim$$

$$+h1\sim)\cdot(a\sim^2+2\cdot a\sim\cdot h1\sim + h0\sim^2)\cdot\sin\Big(\sqrt{h0\sim^2-h1\sim^2}\cdot t\sim\Big) + 2\cdot(a\sim^2 - 2\cdot a\sim\cdot h1\sim$$

$$+h0\sim^2)\cdot(-h1\sim+a\sim)\cdot\sqrt{h0\sim^2-h1\sim^2}\cdot\Big(\frac{1}{4}\Big(3\cdot(h0\sim^2+8\cdot h1\sim^2)\cdot\Big(a\sim^2\cdot h1\sim$$

$$+\Big(-\frac{h0\sim^2}{2}+\frac{5\cdot h1\sim^2}{2}\Big)\cdot a\sim + h1\sim^3\Big)\cdot\Big(a\sim^2+a\sim\cdot h1\sim+\frac{1}{4}\cdot h0\sim^2\Big)$$

$$\cdot e^{-2\cdot t\sim\cdot(a\sim+h1\sim)}\Big) + (h0\sim-h1\sim)\cdot(h0\sim+h1\sim)\cdot\Big(a\sim^2 - a\sim\cdot h1\sim + \frac{1}{4}\cdot h0\sim^2\Big)$$

$$\cdot\Big(a\sim^3 + 4\cdot a\sim^2\cdot h1\sim + \Big(\frac{5\cdot h0\sim^2}{8}+5\cdot h1\sim^2\Big)\cdot a\sim + \frac{h0\sim^2\cdot h1\sim}{4} + 2\cdot h1\sim^3\Big)\Big)\Big)\cdot s\sim^6\Big)$$

$$\Big/ \Big((h0\sim^2-h1\sim^2)^{3/2}\cdot(16\cdot a\sim^4+8\cdot a\sim^2\cdot h0\sim^2 - 16\cdot a\sim^2\cdot h1\sim^2 + h0\sim^4)\cdot(a\sim^2$$

$$+2\cdot a\sim\cdot h1\sim+h0\sim^2)\cdot(h0\sim^2+8\cdot h1\sim^2)\cdot(a\sim^2-2\cdot a\sim\cdot h1\sim+h0\sim^2)\cdot(a\sim^2$$

$$-h1\sim^2)\Big)$$

To calculate the moments of the 4th order, we introduce a characteristic function for a 8-dimensional normal system (under the identifier E), and the elements k_{ij} of the corresponding correlation matrix are calculated using Kz.

```
> u:=array(1..8):k:=array(1..8,1..8):
> E:=unapply(exp(-(1/2)*sum(sum(k[i,j2]*u[i]*u[j2],i=1..8),j2=
  1..8)),k,u):
> for j from 1 to 8 do k[j,j]:=k[1,1] od:
> DE:=unapply(diff(E(k,u),u[1],u[2],u[3],u[4],u[5],u[6],u[7],u
  [8]),k,u):
```

Using E(k,u), we obtain:

$$\mu_4(X) = \frac{1}{i^8}\frac{\partial^8}{\partial u_1\,\partial u_2\,\partial u_3\,\partial u_4\,\partial u_5\,\partial u_6\,\partial u_7\,\partial u_8}E(k,u)\Big|_{u_1=u_2=u_3=u_4=u_5=u_6=u_7=u_8=0}.$$

```
> DE0:=unapply(subs({u[1]=0,u[2]=0,u[3]=0,u[4]=0,u[5]=0,u[6]=
  0,u[7]=0,u[8]=0},DE(k,u)),k):
> mu4x:=unapply(simplify(subs({k[1,2]=k[1,1],k[2,1]=k[1,1],k
  [3,4]=k[1,1],k[4,3]=k[1,1],k[5,6]=k[1,1],k[6,5]=k[1,1],k[7,
  8]=k[1,1],k[8,7]=k[1,1],
                                k[3,1]=k[1,3],k[1,4]=k[1,3],k[4,1]=k
  [1,3],k[2,3]=k[1,3],k[3,2]=k[1,3],k[2,4]=k[1,3],k[4,2]=k[1,
  3],
                         k[5,1]=k[1,5],k[1,6]=k[1,5],k[6,1]=k[1,5],k
  [2,5]=k[1,5],k[5,2]=k[1,5],k[2,6]=k[1,5],k[6,2]=k[1,5],

          k[7,1]=k[1,7],k[1,8]=k[1,7],k[8,1]=k[1,7],k[2,7]=k[1,
  7],k[7,2]=k[1,7],k[2,8]=k[1,7],k[8,2]=k[1,7],
                                                                    k
  [5,3]=k[3,5],k[4,5]=k[3,5],k[5,4]=k[3,5],k[3,6]=k[3,5],k[6,
  3]=k[3,5],k[4,6]=k[3,5],k[6,4]=k[3,5],
                                                          k[7,3]=k
  [3,7],k[4,7]=k[3,7],k[7,4]=k[3,7],k[3,8]=k[3,7],k[8,3]=k[3,
  7],k[4,8]=k[3,7],k[8,4]=k[3,7],

                                                    k[7,5]=k[5,7],k
  [5,8]=k[5,7],k[8,5]=k[5,7],k[6,7]=k[5,7],k[7,6]=k[5,7],k[6,
  8]=k[5,7],k[8,6]=k[5,7]},DE0(k))),k);
```

$$mu4x := k \mapsto k_{1,1}^4 + \left(2\cdot k_{1,3}^2 + 2\cdot k_{1,5}^2 + 2\cdot k_{1,7}^2 + 2\cdot k_{3,5}^2 + 2\cdot k_{3,7}^2 + 2\cdot k_{5,7}^2\right)\cdot k_{1,1}^2 + \left(\left(8\cdot k_{1,5}\cdot k_{3,5} + 8\cdot k_{1,7}\cdot k_{3,7}\right)\cdot k_{1,3} + 8\cdot k_{5,7}\cdot\left(k_{1,5}\cdot k_{1,7} + k_{3,5}\cdot k_{3,7}\right)\right)\cdot k_{1,1} + 4\cdot k_{1,3}^2\cdot k_{5,7}^2 + 16\cdot k_{5,7}\cdot\left(k_{1,5}\cdot k_{3,7} + k_{1,7}\cdot k_{3,5}\right)\cdot k_{1,3} + 4\cdot k_{1,5}^2\cdot k_{3,7}^2 + 16\cdot k_{1,5}\cdot k_{1,7}\cdot k_{3,7}\cdot k_{3,5} + 4\cdot k_{1,7}^2\cdot k_{3,5}^2 \quad (21)$$

```
> k[1,1]:=s^2:
> k[1,3]:=Kz(t1,t2,s,a):
> k[1,5]:=Kz(t1,t3,s,a):
> k[1,7]:=Kz(t1,t4,s,a):
> k[3,5]:=Kz(t2,t3,s,a):
> k[3,7]:=Kz(t2,t4,s,a):
> k[5,7]:=Kz(t3,t4,s,a):
> mu4:=unapply(mu4x(k),t1,t2,t3,t4,s,a);
```

$$\mu 4 := (t1, t2, t3, t4, s\sim, a\sim) \mapsto s\sim^8 + \Big(2\cdot s\sim^4\cdot\left(e^{-a\sim\cdot|t1-t2|}\right)^2 + 2\cdot s\sim^4\cdot\left(e^{-a\sim\cdot|t1-t3|}\right)^2 + 2\cdot s\sim^4\cdot\left(e^{-a\sim\cdot|t1-t4|}\right)^2 + 2\cdot s\sim^4\cdot\left(e^{-a\sim\cdot|t2-t3|}\right)^2 + 2\cdot s\sim^4\cdot\left(e^{-a\sim\cdot|t2-t4|}\right)^2 + 2\cdot s\sim^4\cdot\left(e^{-a\sim\cdot|t3-t4|}\right)^2\Big)\cdot s\sim^4 + \Big(\big(8\cdot s\sim^4\cdot e^{-a\sim\cdot|t1-t3|}\cdot e^{-a\sim\cdot|t2-t3|} + 8\cdot s\sim^4\cdot e^{-a\sim\cdot|t1-t4|}\cdot e^{-a\sim\cdot|t2-t4|}\big)\cdot s\sim^2\cdot e^{-a\sim\cdot|t1-t2|} + 8\cdot s\sim^2\cdot e^{-a\sim\cdot|t3-t4|}\cdot\big(s\sim^4\cdot e^{-a\sim\cdot|t1-t3|}\cdot e^{-a\sim\cdot|t1-t4|} + s\sim^4\cdot e^{-a\sim\cdot|t2-t3|}\cdot e^{-a\sim\cdot|t2-t4|}\big)\Big)\cdot s\sim^2 + 4\cdot s\sim^8\cdot\left(e^{-a\sim\cdot|t1-t2|}\right)^2\cdot\left(e^{-a\sim\cdot|t3-t4|}\right)^2 + 16\cdot s\sim^4\cdot e^{-a\sim\cdot|t3-t4|}\cdot\big(s\sim^4\cdot e^{-a\sim\cdot|t1-t3|}\cdot e^{-a\sim\cdot|t2-t4|} + s\sim^4\cdot e^{-a\sim\cdot|t1-t4|}\cdot e^{-a\sim\cdot|t2-t3|}\big)\cdot e^{-a\sim\cdot|t1-t2|} + 4\cdot s\sim^8\cdot\left(e^{-a\sim\cdot|t1-t3|}\right)^2\cdot\left(e^{-a\sim\cdot|t2-t4|}\right)^2 + 16\cdot s\sim^8\cdot e^{-a\sim\cdot|t1-t4|}\cdot e^{-a\sim\cdot|t1-t3|}\cdot e^{-a\sim\cdot|t2-t3|}\cdot e^{-a\sim\cdot|t2-t4|} + 4\cdot s\sim^8\cdot\left(e^{-a\sim\cdot|t1-t4|}\right)^2\cdot\left(e^{-a\sim\cdot|t2-t3|}\right)^2 \quad (22)$$

Raising the ratio (L) by 4th power and turning the result into a multiple integral, after finding the mathematical expectation, we get:

$$m_4(\mathrm{Y}) = \int_0^t\int_0^t\int_0^t\int_0^t l(t-t_1)\,l(t-t_2)\,l(t-t_3)\,l(t-t_4)\,mu4(t_1, t_2, t_3, t_4, s, a)\,\mathrm{d}t_1\,\mathrm{d}t_2\,\mathrm{d}t_3\mathrm{d}t_4 \quad (L4)$$

According to the formula (L4), the initial moments of the 4th order under the identifier m4, m4r and m4s are calculated (for a quick calculation of m4r and m4s, we perform fragmentation according to the formula (21)). Next, we calculate the 4th-order central moments under the names mu4y, mu4yr and mu4ys, respectively, at the same time, due to the bulkiness, the expression obtained for mu4ys is not displayed on the display.

```
> Q:=unapply(exp(-a*abs(x)),x,a);
```

$$Q := (x, a\sim) \mapsto e^{-a\sim\cdot|x|} \quad (23)$$

```
> R:=unapply(1+2*(Q(t1-t2,2*a)+Q(t1-t3,2*a)+Q(t1-t4,2*a)+Q(t2-
  t3,2*a)+Q(t2-t4,2*a)+Q(t3-t4,2*a))+8*(Q(t1-t3,a)*Q(t2-t3,a)+
  Q(t1-t4,a)*Q(t2-t4,a))*Q(t1-t2,a)+8*Q(t3-t4,a)*(Q(t1-t3,a)*Q
  (t1-t4,a)+Q(t2-t3,a)*Q(t2-t4,a))+4*Q(t1-t2,2*a)*Q(t3-t4,2*a)
  +16*Q(t3-t4,a)*(Q(t1-t3,a)*Q(t2-t4,a)+Q(t1-t4,a)*Q(t2-t3,a))
  *Q(t1-t2,a)+4*Q(t1-t3,2*a)*Q(t2-t4,2*a)+16*Q(t1-t3,a)*Q(t1-
  t4,a)*Q(t2-t3,a)*Q(t2-t4,a)+4*Q(t1-t4,2*a)*Q(t2-t3,2*a),t1,
  t2,t3,t4,a):
> m4n:=unapply(int(int(R(t1,t2,t3,t4,a),t1=0..t),t2=0..t),t,
  t3,t4,a):
> m4:=unapply(s^8*int(int(m4n(t,t3,t4,a),t3=0..t),t4=0..t),t,
  a,s):
> m4r1n:=unapply(int(int(int(int(l(t-t1,rho)*l(t-t2,rho)*l(t-
  t3,rho)*l(t-t4,rho),t1=0..t),t2=0..t),t3=0..t),t4=0..t),t,
  rho):
> m4r2:=unapply(int(int(int(int(l(t-t1,rho)*l(t-t2,rho)*l(t-
  t3,rho)*l(t-t4,rho)*2*Q(t1-t2,2*a),t1=0..t),t2=0..t),t3=0..
  t),t4=0..t),t,a,rho):
> m4r2n:=unapply(6*m4r2(t,a,rho),t,a,rho):
> m4r3:=unapply(int(int(l(t-t1,rho)*l(t-t2,rho)*8*Q(t1-t2,a)*Q
  (t1-t3,a)*Q(t2-t3,a),t1=0..t),t2=0..t),t,t3,t4,a,rho):
> m4r3n:=unapply(4*int(int(l(t-t3,rho)*l(t-t4,rho)*m4r3(t,t3,
  t4,a,rho),t3=0..t),t4=0..t),t,a,rho):
> m4r4:=unapply(int(int(l(t-t1,rho)*l(t-t2,rho)*4*Q(t1-t2,2*a)
  *Q(t3-t4,2*a),t1=0..t),t2=0..t),t,t3,t4,a,rho):
```

```
> m4r4n:=unapply(3*int(int(l(t-t3,rho)*l(t-t4,rho)*m4r4(t,t3,
  t4,a,rho),t3=0..t),t4=0..t),t,a,rho):
> m4r5:=unapply(int(int(l(t-t1,rho)*l(t-t2,rho)*16*Q(t3-t4,a)*
  Q(t1-t3,a)*Q(t2-t4,a)*Q(t1-t2,a),t1=0..t),t2=0..t),t,t3,t4,
  a,rho):
> m4r5n:=unapply(3*int(int(l(t-t3,rho)*l(t-t4,rho)*m4r5(t,t3,
  t4,a,rho),t3=0..t),t4=0..t),t,a,rho):
> m4r:=unapply(s^8*simplify(m4r1n(t,rho)+m4r2n(t,a,rho)+m4r3n
  (t,a,rho)+m4r4n(t,a,rho)+m4r5n(t,a,rho)),t,a,s,rho):
> mu4y:=unapply(simplify(m4(t,a,s)+6*m1(t,s)^2*m2(t,a,s)-4*m1
  (t,s)*m3(t,a,s)-3*m1(t,s)^4),t,a,s);
```

$$mu4y := (t\sim, a\sim, s\sim) \mapsto \frac{1}{a\sim^4}\Big(3\cdot s\sim^8\cdot\big(32\cdot t\sim^2\cdot a\sim^2\cdot e^{-2\cdot t\sim\cdot a\sim} + 4\cdot t\sim^2\cdot a\sim^2 + 84\cdot t\sim \cdot e^{-2\cdot t\sim\cdot a\sim}\cdot a\sim + 3\cdot e^{-4\cdot t\sim\cdot a\sim} + 36\cdot t\sim\cdot a\sim + 54\cdot e^{-2\cdot t\sim\cdot a\sim} - 57\big)\Big) \quad \textbf{(24)}$$

```
> mu4yr:=unapply(simplify(m4r(t,a,s,rho)+6*m1r(t,s,rho)^2*m2r
  (t,a,s,rho)-4*m1r(t,s,rho)*m3r(t,a,s,rho)-3*m1r(t,s,rho)^4),
  t,a,s,rho);
```

$$mu4yr := (t\sim, a\sim, s\sim, \rho\sim) \mapsto \Bigg(6144\cdot\Bigg((a\sim-\rho\sim)\cdot\left(a\sim^2-\frac{5}{16}\cdot a\sim\cdot\rho\sim-\frac{15}{32}\cdot\rho\sim^2\right) \quad \textbf{(25)}$$

$$\cdot\left(a\sim+\frac{\rho\sim}{2}\right)\cdot(a\sim+\rho\sim)\cdot\left(a\sim-\frac{3\cdot\rho\sim}{2}\right)\cdot e^{-(3\cdot\rho\sim+2\cdot a\sim)\cdot t\sim}+\left(a\sim^2+\frac{5}{16}\cdot a\sim\cdot\rho\sim\right.$$

$$\left.-\frac{15}{32}\cdot\rho\sim^2\right)\cdot(a\sim-\rho\sim)\cdot\left(a\sim-\frac{\rho\sim}{2}\right)\cdot\left(a\sim+\frac{3\cdot\rho\sim}{2}\right)\cdot(a\sim+\rho\sim)$$

$$\cdot e^{-t\sim\cdot(2\cdot a\sim+\rho\sim)}+\left(\frac{27}{128}\cdot\rho\sim^6+\frac{3}{32}\cdot a\sim^4\cdot\rho\sim^2-\frac{39}{128}\cdot a\sim^2\cdot\rho\sim^4\right)$$

$$\cdot e^{-2\cdot t\sim\cdot(2\cdot a\sim+\rho\sim)}+\left(-2\cdot a\sim^6+\frac{9}{32}\cdot\rho\sim^6+\frac{11}{2}\cdot a\sim^4\cdot\rho\sim^2-\frac{19}{8}\cdot a\sim^2\cdot\rho\sim^4\right)$$

$$\cdot e^{-2\cdot t\sim\cdot(a\sim+\rho\sim)}$$

$$+\frac{1}{32}\Bigg(\left(a\sim^2-\frac{25}{2}\cdot a\sim\cdot\rho\sim+\frac{15}{2}\cdot\rho\sim^2\right)\cdot\left(a\sim+\frac{3\cdot\rho\sim}{2}\right)\cdot\left(a\sim+\frac{\rho\sim}{2}\right)^2\cdot(a\sim$$

$$+\rho\sim)\cdot e^{-4\cdot\rho\sim\cdot t\sim}\Bigg)-\frac{1}{16}\Bigg((a\sim-\rho\sim)\cdot\Bigg(\left(a\sim+\frac{3\cdot\rho\sim}{2}\right)\cdot\left(a\sim+\frac{\rho\sim}{2}\right)\cdot(a\sim$$

$$+\rho\sim)\cdot e^{-2\cdot\rho\sim\cdot t\sim}-\frac{a\sim^3}{2}-6\cdot a\sim^2\cdot\rho\sim-\frac{5\cdot a\sim\cdot\rho\sim^2}{8}+\frac{15\cdot\rho\sim^3}{8}\Bigg)\cdot\left(a\sim-\frac{\rho\sim}{2}\right)\cdot\Big(a\sim$$

$$-\frac{3\cdot\rho\sim}{2}\Big)\Big)\Big)\cdot s\sim^{8}\Big) \Big/ \Big(64\cdot a\sim^{8}\cdot\rho\sim^{2}-240\cdot a\sim^{6}\cdot\rho\sim^{4}+252\cdot a\sim^{4}\cdot\rho\sim^{6}-85\cdot a\sim^{2}\cdot\rho\sim^{8}+9\cdot\rho\sim^{10}\Big)$$

```
> m4s1n:=unapply(int(int(int(int(ls(t-t1,h0,h1)*ls(t-t2,h0,h1)
  *ls(t-t3,h0,h1)*ls(t-t4,h0,h1),t1=0..t),t2=0..t),t3=0..t),
  t4=0..t),t,h0,h1):
> m4s2n:=unapply(6*int(int(int(int(ls(t-t1,h0,h1)*ls(t-t2,h0,
  h1)*ls(t-t3,h0,h1)*ls(t-t4,h0,h1)*2*Q(t1-t2,2*a),t1=0..t),
  t2=0..t),t3=0..t),t4=0..t),t,a,h0,h1):
> m4s3:=unapply(int(int(ls(t-t1,h0,h1)*ls(t-t2,h0,h1)*8*Q(t1-
  t2,a)*Q(t1-t3,a)*Q(t2-t3,a),t1=0..t),t2=0..t),t,t3,t4,a,h0,
  h1):
> m4s3n:=unapply(4*int(int(ls(t-t3,h0,h1)*ls(t-t4,h0,h1)*m4s3
  (t,t3,t4,a,h0,h1),t3=0..t),t4=0..t),t,a,h0,h1):
> m4s4:=unapply(int(int(ls(t-t1,h0,h1)*ls(t-t2,h0,h1)*4*Q(t1-
  t2,2*a)*Q(t3-t4,2*a),t1=0..t),t2=0..t),t,t3,t4,a,h0,h1):
> m4s4n:=unapply(3*int(int(ls(t-t3,h0,h1)*ls(t-t4,h0,h1)*m4s4
  (t,t3,t4,a,h0,h1),t3=0..t),t4=0..t),t,a,h0,h1):
> m4s5:=unapply(16*int(int(ls(t-t1,h0,h1)*ls(t-t2,h0,h1)*Q(t3-
  t4,a)*Q(t1-t3,a)*Q(t2-t4,a)*Q(t1-t2,a),t1=0..t),t2=0..t),t,
  t3,t4,h0,h1):
> m4s5n:=unapply(3*int(int(ls(t-t3,h0,h1)*ls(t-t4,h0,h1),t3=0.
  .t),t4=0..t),t,a,h0,h1):
> m4s:=unapply(s^8*simplify(m4s1n(t,h0,h1)+m4s2n(t,a,h0,h1)+
  m4s3n(t,a,h0,h1)+m4s4n(t,a,h0,h1)+m4s5n(t,a,h0,h1)),t,a,s,
  h0,h1):
> mu4ys:=unapply(simplify(m4s(t,a,s,h0,h1)+6*m1s(t,s,h0,h1)^2*
  m2s(t,a,s,h0,h1)-4*m1s(t,s,h0,h1)*m3sn(t,a,s,h0,h1)-3*m1s(t,
  s,h0,h1)^4),t,a,s,h0,h1):
```

We introduce the density of the standard normal distribution and its 3rd and 4th derivatives:

```
> phi:=unapply((1/sqrt(2*Pi))*exp(-x^2/2),x);
```

$$\phi := x \mapsto \frac{\sqrt{2}\cdot e^{-\frac{x^2}{2}}}{2\cdot\sqrt{\pi}} \tag{26}$$

```
> phi3:=unapply(diff(phi(x),x$3),x):
> phi4:=unapply(diff(phi(x),x$4),x):
```

We define Sk(Y) and Ex(Y), and based on the formula (H), we define densities under the names gy, gyr, gys (due to the bulkiness, the expression for gys is not displayed) and at t=1/a, 2/a, 100/a we plot the corresponding densities, and for comparison in the same coordinate system we set density graph of the standard normal distribution:

```
> Sky:=unapply(mu3y(t,a,s)/mu2y(t,a,s)^(3/2),t,a,s):
> Exy:=unapply(mu4y(t,a,s)/mu2y(t,a,s)^2-3,t,a,s):
> gy:=unapply(simplify(phi(y)-(1/3!)*Sky(t,a,s)*phi3(y)+(1/4!)
  *Exy(t,a,s)*phi4(y)),y);
```

$$gy := y \mapsto \frac{1}{\sqrt{\pi}\cdot\left(2\cdot t\sim\cdot a\sim+e^{-2\cdot t\sim\cdot a\sim}-1\right)^{5/2}}\Bigg(2\cdot\sqrt{2}\cdot\Big(\Big(\Big(\Big(\Big(t\sim^{2}\cdot a\sim^{2}+\frac{5}{2}\cdot t\sim\cdot a\sim \tag{27}$$

$$+\frac{7}{4}\Big)\cdot y^4+\Big(-6\cdot t\sim^2\cdot a\sim^2-15\cdot t\sim\cdot a\sim-\frac{21}{2}\Big)\cdot y^2+3\cdot t\sim^2\cdot a\sim^2+\frac{17\cdot t\sim\cdot a\sim}{2}$$

$$+\frac{19}{4}\Big)\cdot \mathrm{e}^{-2\cdot t\sim\cdot a\sim}+\Big(-\frac{3}{8}\cdot y^2+\frac{1}{16}\cdot y^4+\frac{7}{16}\Big)\cdot \mathrm{e}^{-4\cdot t\sim\cdot a\sim}+\Big(\frac{5\cdot t\sim\cdot a\sim}{4}-\frac{29}{16}\Big)$$

$$\cdot y^4+\Big(-\frac{15\cdot t\sim\cdot a\sim}{2}+\frac{87}{8}\Big)\cdot y^2-\frac{83}{16}+t\sim^2\cdot a\sim^2+\frac{11\cdot t\sim\cdot a\sim}{4}\Big)$$

$$\cdot\sqrt{2\cdot t\sim\cdot a\sim+\mathrm{e}^{-2\cdot t\sim\cdot a\sim}-1}+\left(y^2-3\right)\cdot y\cdot\Big(\left(t\sim^2\cdot a\sim^2+t\sim\cdot a\sim-1\right)\cdot \mathrm{e}^{-2\cdot t\sim\cdot a\sim}$$

$$+\Big(\frac{t\sim\cdot a\sim}{2}+\frac{1}{2}\Big)\cdot \mathrm{e}^{-4\cdot t\sim\cdot a\sim}+\frac{1}{2}+t\sim^2\cdot a\sim^2-\frac{3\cdot t\sim\cdot a\sim}{2}\Big)\Big)\cdot \mathrm{e}^{-\frac{y^2}{2}}\Big)$$

```
> gy1:=unapply(subs({t=1/a,s=1},gy(y)),y):
> gy2:=unapply(subs({t=2/a,s=1},gy(y)),y):
> gy100:=unapply(subs({t=100/a,s=1},gy(y)),y):
> plot([gy1(y),gy2(y),gy100(y),phi(y)],y=-5..5,style=[line,
  line,line,point],color=[green,blue,red,black]);
```




```
> evalf(int(gy100(y),y=-infinity..infinity));
```

$$1.000000000 \quad (28)$$

```
> Skyr:=unapply(mu3yr(t,a,s,rho)/mu2yr(t,a,s,rho)^(3/2),t,a,s,
  rho):
> Exyr:=unapply(mu4yr(t,a,s,rho)/mu2yr(t,a,s,rho)^2-3,t,a,s,
  rho):
> gyr:=unapply(phi(y)-(1/3!)*phi3(y)*Skyr(t,a,s,rho)+(1/4!)*
  phi4(y)*Exyr(t,a,s,rho),y,t,a,s,rho);
```

$$gyr := (y, t\sim, a\sim, s\sim, \rho\sim) \mapsto \frac{\sqrt{2}\cdot e^{-\frac{y^2}{2}}}{2\cdot\sqrt{\pi}} + \Bigg(8\cdot\Bigg(\frac{3\cdot\sqrt{2}\cdot y\cdot e^{-\frac{y^2}{2}}}{2\cdot\sqrt{\pi}} - \frac{\sqrt{2}\cdot y^3\cdot e^{-\frac{y^2}{2}}}{2\cdot\sqrt{\pi}}\Bigg) \qquad \textbf{(29)}$$

$$\cdot\Bigg(-3\cdot(a\sim+\rho\sim)\cdot\Big(a\sim-\frac{\rho\sim}{2}\Big)\cdot e^{-t\sim\cdot(2\cdot a\sim+\rho\sim)} + 3\cdot\Big(a\sim+\frac{\rho\sim}{2}\Big)\cdot(a\sim-\rho\sim)$$

$$\cdot e^{-2\cdot t\sim\cdot(a\sim+\rho\sim)} + (a\sim+\rho\sim)\cdot\Big(a\sim+\frac{\rho\sim}{2}\Big)\cdot e^{-3\cdot t\sim\cdot\rho\sim} - (a\sim-\rho\sim)\cdot\Big(a\sim-\frac{\rho\sim}{2}\Big)\Bigg)\Bigg)$$

$$\cdot s\sim^6\Bigg) \Bigg/ \Bigg(3\cdot(4\cdot a\sim^4\cdot\rho\sim - 5\cdot a\sim^2\cdot\rho\sim^3 + \rho\sim^5)\cdot\Bigg($$

$$-\frac{4\cdot s\sim^4\cdot\Big(-e^{-t\sim\cdot(2\cdot a\sim+\rho\sim)}\cdot\rho\sim + \Big(a\sim+\frac{\rho\sim}{2}\Big)\cdot e^{-2\cdot t\sim\cdot\rho\sim} - a\sim + \frac{\rho\sim}{2}\Big)}{4\cdot a\sim^2\cdot\rho\sim - \rho\sim^3}\Bigg)^{3/}$$

$$^{2}\Bigg) + \frac{1}{24}\Bigg(\Bigg(\Bigg(\frac{3\cdot\sqrt{2}\cdot e^{-\frac{y^2}{2}}}{2\cdot\sqrt{\pi}} - \frac{3\cdot\sqrt{2}\cdot y^2\cdot e^{-\frac{y^2}{2}}}{\sqrt{\pi}} + \frac{\sqrt{2}\cdot y^4\cdot e^{-\frac{y^2}{2}}}{2\cdot\sqrt{\pi}}\Bigg)\cdot\Bigg(\Bigg(\Bigg(384$$

$$\cdot\Bigg((a\sim-\rho\sim)\cdot\Big(a\sim^2 - \frac{5}{16}\cdot\rho\sim\cdot a\sim - \frac{15}{32}\cdot\rho\sim^2\Big)\cdot\Big(a\sim+\frac{\rho\sim}{2}\Big)\cdot(a\sim+\rho\sim)\cdot\Big(a\sim$$

$$-\frac{3\cdot\rho\sim}{2}\Big)\cdot e^{-(3\cdot\rho\sim+2\cdot a\sim)\cdot t\sim} + \Big(a\sim^2 + \frac{5}{16}\cdot\rho\sim\cdot a\sim - \frac{15}{32}\cdot\rho\sim^2\Big)\cdot(a\sim-\rho\sim)\cdot\Big(a\sim$$

$$-\frac{\rho\sim}{2}\Big)\cdot\Big(a\sim+\frac{3\cdot\rho\sim}{2}\Big)\cdot(a\sim+\rho\sim)\cdot e^{-t\sim\cdot(2\cdot a\sim+\rho\sim)} + \Big(\frac{27}{128}\cdot\rho\sim^6 + \frac{3}{32}\cdot a\sim^4$$

$$\cdot\rho\sim^2 - \frac{39}{128}\cdot a\sim^2\cdot\rho\sim^4\Big)\cdot e^{-2\cdot t\sim\cdot(2\cdot a\sim+\rho\sim)} + \Big(-2\cdot a\sim^6 + \frac{9}{32}\cdot\rho\sim^6 + \frac{11}{2}\cdot a\sim^4\cdot\rho\sim^2$$

$$-\frac{19}{8}\cdot a\sim^2\cdot\rho\sim^4\Big)\cdot e^{-2\cdot t\sim\cdot(a\sim+\rho\sim)}$$

$$+\frac{1}{32}\Bigg(\Big(a\sim^2 - \frac{25}{2}\cdot\rho\sim\cdot a\sim + \frac{15}{2}\cdot\rho\sim^2\Big)\cdot\Big(a\sim+\frac{3\cdot\rho\sim}{2}\Big)\cdot\Big(a\sim+\frac{\rho\sim}{2}\Big)^2\cdot(a\sim$$

$$+\rho\sim)\cdot e^{-4\cdot t\sim\cdot\rho\sim}\Bigg) - \frac{1}{16}\Bigg((a\sim-\rho\sim)\cdot\Bigg(\Big(a\sim+\frac{3\cdot\rho\sim}{2}\Big)\cdot\Big(a\sim+\frac{\rho\sim}{2}\Big)\cdot(a\sim$$

$$+\rho{\sim}\big)\cdot \mathrm{e}^{-2\cdot t{\sim}\cdot\rho{\sim}} - \frac{a{\sim}^3}{2} - 6\cdot a{\sim}^2\cdot\rho{\sim} - \frac{5\cdot a{\sim}\cdot\rho{\sim}^2}{8} + \frac{15\cdot\rho{\sim}^3}{8}\Bigg)\cdot\left(a{\sim} - \frac{\rho{\sim}}{2}\right)\cdot\Bigg(a{\sim}$$

$$-\frac{3\cdot\rho{\sim}}{2}\Bigg)\Bigg)\Bigg)\Bigg)\cdot\left(4\cdot a{\sim}^2\cdot\rho{\sim} - \rho{\sim}^3\right)^2\Bigg) \Bigg/ \Bigg(\Big(64\cdot a{\sim}^8\cdot\rho{\sim}^2 - 240\cdot a{\sim}^6\cdot\rho{\sim}^4 + 252\cdot a{\sim}^4$$

$$\cdot\rho{\sim}^6 - 85\cdot a{\sim}^2\cdot\rho{\sim}^8 + 9\cdot\rho{\sim}^{10}\Big)\cdot\Bigg(-\mathrm{e}^{-t{\sim}\cdot(2\cdot a{\sim}+\rho{\sim})}\cdot\rho{\sim} + \left(a{\sim} + \frac{\rho{\sim}}{2}\right)$$

$$\cdot\mathrm{e}^{-2\cdot t{\sim}\cdot\rho{\sim}} - a{\sim} + \frac{\rho{\sim}}{2}\Bigg)^2\Bigg) - 3\Bigg)\Bigg)$$

```
> gyr1:=unapply(subs({a=1,t=1,s=1,rho=0.01},gyr(y,t,a,s,rho)),
  y):
> gyr2:=unapply(subs({a=1,t=2,s=1,rho=0.01},gyr(y,t,a,s,rho)),
  y):
> gyr100:=unapply(subs({a=1,t=100,s=1,rho=0.01},gyr(y,t,a,s,
  rho)),y):
> gyrn:=unapply(subs({a=1,t=100,s=1,rho=3},gyr(y,t,a,s,rho)),
  y):
> plot([gyrn(y),gyr1(y),gyr2(y),gyr100(y),phi(y)],y=-5..5,
  style=[line,line,line,point,point],color=[orange,green,blue,
  red,black]);
```




```
> int(gyr100(y),y=-infinity..infinity);
```

$$1.000000000 \quad (30)$$

```
> Skys:=unapply(mu3ys(t,a,s,h0,h1)/mu2ys(t,a,s,h0,h1)^(3/2),t,
  a,s,h0,h1):
> Exys:=unapply(mu4ys(t,a,s,h0,h1)/mu2ys(t,a,s,h0,h1)^2-3,t,a,
  s,h0,h1):
> gys:=unapply(phi(y)-(1/3!)*phi3(y)*Skys(t,a,s,h0,h1)+(1/4!)*
  phi4(y)*Exys(t,a,s,h0,h1),y,t,a,s,h0,h1):
```

```
> gys100:=unapply(subs({a=1,t=100,s=1,h0=5,h1=3},gys(y,t,a,s,
  h0,h1)),y);
```

(31)

```
> plot([gys100(y),phi(y)],y=-5..5,style=[point,point],color=
  [red,black]);
```



The obtained figures allow us to draw practically useful conclusions: for the first two types of weight functions, an almost normal distribution is obtained for sufficiently large time values, but for the third type of weight functions, the difference from the normal distribution law is quite significant.

References

1. Sveshnikov, A. A. (2014). *Applied methods of the theory of random functions* (Vol. 89). Elsevier.
2. Vrbik, J., & Vrbik, P. (2012). *Informal introduction to stochastic processes with Maple*. Springer Science & Business Media.
3. Yaglom, A. M. (1987). *Correlation theory of stationary and related random functions, Volume I: Basic results* (Vol. 1). New York: Springer.